DELAY PROGRAMS

TIME DELAY USING ONE REGISTER

Label	Opcode	Operand	Comments	T-states
	MVI	C,FFH	;Load register C	7
LOOP:	DCR	C	;Decrement C	4
	JNZ	LOOP	;Jump back to ; decrement C	10/7

Clock frequency of the system f = 2 MHz
Clock period $T = 1/f = 1/2 \times 10^{-6} = 0.5\ \mu s$
Time to execute MVI = 7 T-states × 0.5
$= 3.5\ \mu s$

The time delay in the loop T_L with 2 MHz clock frequency is calculated as

$$TL = (T \times \text{Loop T-states} \times N10)$$

where T_L = Time delay in the loop
T = System clock period
N_{10} = Equivalent decimal number of the hexadecimal count loaded in the delay register

$$T_L = (0.5 \times 10^{-6} \times 14 \times 255)$$
$$= 1785\ \mu s$$
$$\approx 1.8\ ms$$

the adjusted loop delay is

$$T_{LA} = T_L - (3\ \text{T-states} \times \text{Clock period})$$
$$= 1785.0\ \mu s - 1.5\ \mu s = 1783.5\ \mu s$$

$$\text{Total Delay} = \begin{matrix}\text{Time to execute instructions} \\ \text{outside loop}\end{matrix} + \begin{matrix}\text{Time to execute} \\ \text{loop instructions}\end{matrix}$$

$$T_D = T_O + T_{LA}$$
$$= (7 \times 0.5\ \mu s) + 1783.5\ \mu s = 1787\ \mu s$$
$$\approx 1.8\ ms$$

TIME DELAY USING A REGISTER PAIR

Label	Opcode	Operand	Comments	T-states
	LXI	B,2384H	;Load BC with 16-bit count	10
LOOP:	DCX	B	;Decrement (BC) by one	6
	MOV	A,C	;Place contents of C in A	4
	ORA	B	;OR (B) with (C) to set Zero flag	4
	JNZ	LOOP	;If result ≠ 0, jump back to LOOP	10/7

$$2384H = 2 \times (16)^3 + 3 \times (16)^2 + 8 \times (16)^1 + 4(16^0)$$
$$= 9092_{10}$$

If the clock period of the system = 0.5 μs, the delay in the loop T_L is

$$T_L = (0.5 \times 24 \times 9092_{10})$$
$$\approx 109 \text{ ms (without adjusting for the last cycle)}$$
$$\text{Total Delay } T_D = 109 \text{ ms} + T_O$$
$$\approx 109 \text{ ms (The instruction LXI adds only 5 μs.)}$$

TIME DELAY USING A LOOP WITHIN A LOOP TECHNIQUE

```
        MVI B,38H     7T
LOOP2:  MVI C,FFH     7T
LOOP1:  DCR C         4T
        JNZ LOOP1     10/7T
        DCR B         4T
        JNZ LOOP2     10/7T
```

The delay in LOOP1 is $T_{L1} = 1783.5\ \mu s$.

$$
\begin{aligned}
T_{L2} &= 56(T_{L1} + 21\ \text{T-states} \times 0.5\ \mu s) \\
&= 56(1783.5\ \mu s + 10.5\ \mu s) \\
&= 100.46\ \text{ms}
\end{aligned}
$$

HEXADECIMAL COUNTER(FF TO 00H) WITH 1ms TIME DELAY BETWEEN COUNTS(ASSUME CLOCK OF 2MHz)

Label	Mnemonics
	MVI B,00H
NEXT:	DCR B
	MVI C.COUNT
DELAY:	DCR C
	JNZ DELAY
	MOV A,B
	OUT PORT#
	JMP NEXT

the time delay T_L in the loop

$$T_L = 14 \text{ T-states} \times T \text{ (Clock period)} \times \text{Count}$$
$$= 14 \times (0.5 \times 10^{-6}) \times \text{Count}$$
$$= (7.0 \times 10^{-6}) \times \text{Count}$$

The delay outside the loop includes the following instructions:

DCR B	4T
MVI C,COUNT	7T
MOV A,B	4T
OUT PORT	10T
JMP	10T
	35 T-states

Delay outside the loop: $T_O = 35 \text{ T-states} \times T$
$$= 35 \times (0.5 \times 10^{-6})$$
$$= 17.5\ \mu s$$

$$\text{Total Time Delay } T_D = T_O + T_L$$
$$1 \text{ ms} = 17.5 \times 10^{-6} + (7.0 \times 10^{-6}) \times \text{Count}$$
$$\text{Count} = \frac{1 \times 10^{-3} - 17.5 \times 10^{-6}}{7.0 \times 10^{-6}} \approx 140_{10}$$

Zero to nine(modulo 10) counter with 1s time delay between counts (ASSUME CLOCK OF 1 MHz)

```
START:    MVI B,00H
          MOV A, B
DSPLAY:   OUT PORT#      10 } T_O
          LXI H,16-Bit   10 }
LOOP:     DCX H           6 }
          MOV A,L         4 }
          ORA H           4 } T_L: 24 T-states
          JNZ LOOP     10/7 }
          INR B           4 }
          MOV A,B         4 }
          CPI 0AH         7 } T_O
          JNZ DSPLAY   10/7 }
          JZ START
```

$$\text{Loop Delay } T_L = 24 \text{ T-states} \times T \times \text{Count}$$

$$1 \text{ second} = 24 \times 1.0 \times 10^{-6} \times \text{Count}$$

$$\text{Count} = \frac{1}{24 \times 10^{-6}} = 41666 = \text{A2C2H}$$

The instructions outside the loop are: OUT, LXI, INR, MOV, CPI, and JNZ (DSPLAY). These instructions require 45 T-states; therefore, the delay count is calculated as follows:

$$\text{Total Delay } T_D = T_O + T_L$$

$$1 \text{ second} = (45 \times 1.0 \times 10^{-6}) + (24 \times 1.0 \times 10^{+6} \times \text{Count})$$

$$\text{Count} \approx 41665$$

Generate a continuous square wave with a period of 500µs, clock period is 325ns and use bit D_0 to output the square wave

```
         MVI D,AA
ROTATE:  MOV A,D
         RLC

         MOV D,A
         ANI 01H

         OUT PORT1

         MVI B,COUNT (7T)

DELAY:   DCR B (4T)
         JNZ DELAY (10/7T)

         JMP ROTATE (10T)
```

1. The number of instructions outside the loop is seven; it includes six instructions before the loop beginning at the symbol ROTATE and the last instruction JMP.

$$\text{Delay outside the Loop: } T_O = 46 \text{ T-states} \times 325 \text{ ns} = 14.95\ \mu s$$

2. The delay loop includes two instructions (DCR and JNZ) with 14 T-states except for the last cycle, which has 11 T-states.

$$\begin{aligned}\text{Loop Delay: } T_L &= 14 \text{ T-states} \times 325 \text{ ns} \times (\text{Count} - 1) + 11 \text{ T-states} \times 325 \text{ ns} \\ &= 4.5\ \mu s\ (\text{Count} - 1) + 3.575\ \mu s\end{aligned}$$

3. The total delay required is 250 µs. Therefore, the count can be calculated as follows:

$$\begin{aligned}T_D &= T_O + T_L \\ 250\ \mu s &= 14.95\ \mu s + 4.5\ \mu s\ (\text{Count} - 1) + 3.575\ \mu s \\ \text{Count} &= 52.4_{10} = 34H\end{aligned}$$

DELAY SUBROUTINE

```
       MVI B,FFH
LOOP:  DCR B
       JNZ LOOP
       RET
```